ESET 349 – Microcontroller Architecture

TABLES

Dr. Muhammad Faeyz Karim

Table

- Table is a way of organizing information.
- It has list of data items

Table example 1 – a simple list

- A list of 10 sampled data (eg., weight of young boys)



Table example 2 – a more complicated table

- A table contains 5 records of student data



Common tasks on table

- 2 common tasks are sorting and searching for data in memory from a list of elements , eg., list of sampled data
- Sorting and searching techniques/algorithms could easily fill several books

A Simple List in Memory

Note: each data, in this case, is 4 bytes in size

	Address	Memory
	…	…
	0x8010	*53*
	0x800C	*61*
	0x8008	*44*
	0x8004	*81*
base address	0x8000	*76*

Addressing data of Table

- Use ARM addressing modes: pre-indexed addressing with an offset to address an element
- Example1: **LDR r6, [r5, r4]**
  r4 contains the **offset address**
- Example 2: **LDR r6, [r5, r4, LSL #2]**
  r4 is the **index**
- Note:
  - r5 stores the base (starting) address
  - r6 will be loaded with the addressed data
  - example 2 accounts for the size of the data by multiplying the index by 4 bytes
  - index starts from 0 to (n – 1)



Accessing a Simple List in Memory



Accessing a simple list of data; each data is half word (2 bytes) in size

- Ex1, LDRH r6, [r5, r4] ;r4 - offset
- Ex2, LDRH r6, [r5, r4, LSL #1] ;r4 - index



x2
(2 bytes)

Accessing a simple list of data; each data is 1 byte in size

- Ex1, LDRB r6, [r5, r4] ;r4 – offset or
- Ex2, LDRB r6, [r5, r4] ;r4 - index

Example 9.1 – arithmetic progression

In mathematics, an arithmetic progression (AP) or arithmetic sequence such as

1, 5, 9, 13, 17

has a constant difference between terms. Example, the first term is a_0(=1), the common difference is d(=4), and the number of terms is n(=5). The explicit formula is $a_i = a_{i-1} + d$

Write an assembly program to create a table of arithmetic progression (AP) in memory. The starting address of the table in memory is 0x8000. Let r0 = n and r1 = a_1 and r2 = d. Assume that the AP are 32-bit data and d is positive.

```
AREA ArithProgression, CODE, READONLY

; Registers used:
; r0 = n = 5
; r1 = AP(i), initial value = a0 = 1
; r2 = d, difference term = 4
; r3 = starting address of table in memory = 0x8000
; r4 = i, loop counter, initially 0
```

Example 9.1 – AP Codes

```
TABLE_BASE EQU 0x00008000    ;base of table
    ENTRY
    MOV  r0, #5      ;n = 5
    MOV  r1, #1      ;a0 = 1
    MOV  r2, #4      ;d = 4
    MOV  r3, #TABLE_BASE
    MOV  r4, #0

    STR    r1, [r3]  ;store AP(0)
```

```
loop   ADD r4, r4, #1    ;i = i + 1
       CMP r4, r0        ;i < n, proceed
       BGE done          ;else finish!

       ADD r1, r1, r2    ;AP(i) = AP(i-1) + d
       STR  r1, [r3, r4, LSL #2]  ;store AP(i)
       B loop

done  B done
       END
```

Jump Table

- Instead of holding data, these tables hold addresses of subroutines.
- Useful for replacing a series of comparisons and branches.
- For example, a controller may receive an input from a keypad, and based on the number that was pressed, jump to a particular subroutine to configure registers, display data, or initialize variables.
- Much like the switch statement in C.

Example 9.2 Jump Table

- function **arithfunc** takes three arguments
- The first argument determines the operation to be performed on the other two arguments
- If the first argument is a “0”, the returned result is the sum of the second and third arguments
- If the first argument is a “1”, the returned result is the difference of the second and third arguments

Example 9.2 Jump Table Codes



```
        AREA Jump, CODE, READONLY
        ENTRY
Start
        MOV  r0, #0 ; option = 0 implies do addition
        MOV  r1, #3 ;2nd argument
        MOV  r2, #2 ;3rd argument
        BL  arithfunc ; call the function
stop    B  stop
```

Example 9.2 Jump Table Codes - cont

```
arithfunc       ; label the function
    ADR  r3, jumpTable   ; load address of jump table
    LDR  pc, [r3, r0, LSL #2]   ; jump to the appropriate routine
jumpTable
    DCD  DoAdd ;r0=0(add), contains address of DoAdd
    DCD  DoSub  ;r0=1(sub), contains address of DoSub
DoAdd
    ADD  r4, r1, r2; operation DoAdd, when r0 = 0
    BX     lr ; return to calling program
DoSub
    SUB  r4, r1, r2  ; operation DoSub, when r0 = 1
    BX     lr       ; return to calling program
    END             ; Mark the end of this file
```


Project
Target 1
Source Group 1
9JTable.s
P...
B...
F...
T...
9JTable.s
01 AREA JumpTable, CODE, READONLY
02 ENTRY
03 start
04 MOV r0, #0
05 MOV r1, #3
06 MOV r2, #2
07 BL arithfunc
08 stop B stop
09 arithfunc
10 ADR r3, jumpTable
11 LDR pc, [r3, r0, LSL #2]
12 jumpTable
13 DCD DoAdd
14 DCD DoSub
15 DoAdd
16 ADD r4, r1, r2
17 BX lr
18 DoSub
19 SUB r4, r1, r2
20 BX lr
Build Output
assembling 9JTable.s...
linking...
Program Size: Code=52 RO-data=0 RW-data=0 ZI-data=0
"9JTable.axf" - 0 Error(s), 0 Warning(s).


Register
Value
Current
R0 0x00000000
R1 0x00000003
R2 0x00000002
R3 0x0000001C
R4 0x00000000
R5 0x00000000
R6 0x00000000
R7 0x00000000
R8 0x00000000
R9 0x00000000
R10 0x00000000
R11 0x00000000
R12 0x00000000
R13 (SP) 0x00000000
R14 (LR) 0x00000010
R15 (PC) 0x00000018
CPSR 0x000000D3
SPSR 0x00000000
Project
Registers
11: LDR pc, [r3, r0,
12: jumpTable
9JTable.s
03 start
04 MOV r0, #0
05 MOV r1, #3
06 MOV r2, #2
07 BL arithfunc
08 stop B stop
09 arithfunc
10 ADR r3, jumpTable
11 LDR pc, [r3, r0, LSL #2]
12 jumpTable
13 DCD DoAdd
14 DCD DoSub
15 DoAdd
16 ADD r4, r1, r2
17 BX lr
18 DoSub
Command
ASSIGN BreakDisable
Memory 1
Address: 0x00
0x00000000: 00 00 A0 E3 03 10 A0 E3 02 20 A0 E3 00 00 00 EB
0x0000001C: 24 00 00 00 2C 00 00 00 02 40 81 E0 1E FF 2F E1


Register
Value
Current
R0 0x00000000
R1 0x00000003
R2 0x00000002
R3 0x0000001C
R4 0x00000000
R5 0x00000000
R6 0x00000000
R7 0x00000000
R8 0x00000000
R9 0x00000000
R10 0x00000000
R11 0x00000000
R12 0x00000000
R13 (SP) 0x00000000
R14 (LR) 0x00000010
R15 (PC) 0x00000024
CPSR 0x000000D3
SPSR 0x00000000
Project
Registers
16: ADD r4, r1, r
0x00000024 E0814002 ADD R4,R1,R2
9JTable.s
03 start
04 MOV r0, #0
05 MOV r1, #3
06 MOV r2, #2
07 BL arithfunc
08 stop B stop
09 arithfunc
10 ADR r3, jumpTable
11 LDR pc, [r3, r0, LSL #2]
12 jumpTable
13 DCD DoAdd
14 DCD DoSub
15 DoAdd
16 ADD r4, r1, r2
17 BX lr
18 DoSub
Command
>
ASSIGN BreakDisable
Memory 1
Address: 0x00
0x00000000: 00 00 A0 E3 03 10 A0 E3 02 20 A0 E3 00 00 0
0x0000001C: 24 00 00 00 2C 00 00 00 02 40 81 E0 1E FF 2


Register
Value
Current
R0 0x00000000
R1 0x00000003
R2 0x00000002
R3 0x0000001C
R4 0x00000005
R5 0x00000000
R6 0x00000000
R7 0x00000000
R8 0x00000000
R9 0x00000000
R10 0x00000000
R11 0x00000000
R12 0x00000000
R13 (SP) 0x00000000
R14 (LR) 0x00000010
R15 (PC) 0x00000010
CPSR 0x000000D3
SPSR 0x00000000
Project
Registers
8: stop B stop
9: arithfunc
9JTable.s
03 start
04 MOV r0, #0
05 MOV r1, #3
06 MOV r2, #2
07 BL arithfunc
08 stop B stop
09 arithfunc
10 ADR r3, jumpTable
11 LDR pc, [r3, r0, LSL #2]
12 jumpTable
13 DCD DoAdd
14 DCD DoSub
15 DoAdd
16 ADD r4, r1, r2
17 BX lr
18 DoSub
Command
ASSIGN BreakDisable
Memory 1
Address: 0x00
0x00000000: 00 00 A0 E3 03 10 A0 E3 02 20 A0 E3 00 00 00
0x0000001C: 24 00 00 00 2C 00 00 00 02 40 81 E0 1E FF 2F

Exercise: Change r0 = 1 and observe the change in operation!

Bubble Sort

- In order to search, we need to sort the table (list) first
- A bubble sort is a simple way to sort entries
- Algorithm
  - Compare 2 entries in a list (entry [j] and entry[j+1])
  - If entry[j] is larger, then swap the entries
  - Repeat this until the last 2 entries are compared
  - The largest element will now be the last entry
  - This is repeated until the smallest element is swapped or bubbled to be the first entry

Example 9.3 Bubble Sort – ARM codes

```
        AREA BubbleSort, CODE, READONLY
num EQU  20
        ENTRY
        MOV  r0, #num    ;n
        SUB   r1, r0, #1 ;n-1
        LDR   r2, =elements   ;base address of table
        MOV  r3, #0    ;loop counter i
```

```
loop1   CMP  r3, r0  ;i < n
        BGE   done     ;done
        MOV r4, #0     ;loop counter j
loop2   CMP  r4, r1;j < (n-1)
        BGE   end2
        ADD  r5, r4, #1;j+1
        LDR   r6, [r2, r4, LSL #2];load entry[j]
        LDR   r7, [r2, r5, LSL #2];load entry[j+1]
        CMP  r6, r7
        BLE   Noswap
        STR   r6, [r2, r5, LSL #2] ;swap if necessary
        STR   r7, [r2, r4, LSL #2]
Noswap
        ADD r4, r4, #1   ;inc j
        B      loop2
end2    ADD r3, r3, #1  ;inc i
        B      loop1

done    B       done
```

```
      AREA  data1, DATA,READWRITE
elements
      DCD   3, 4, 2, 54, 6, 33, -5, 2, 1, 0
      DCD   77, 1, 2, 3, 0, -5, -35, 3, 4, 1
END
```





Example 9.3 Bubble Sort – initial file





Search for data in a table

- It is common to search a table for information, using a key
- Once the key matches an item in a table, the address of the matched item is noted and further processing can be done
- The simplest way is to search the table from the beginning to the end of the table → **sequential search**
- Another more efficient way is to use **binary search,** which requires the table to be sorted, either in ascending or descending order

Example 9.4 Sequential Search – ARM codes 1



```
    AREA SequentialSearch, CODE, READONLY
    ENTRY
    MOV    r0, #0;found = 0 -> not found
    MOV    r1, #5;n = 5 -> number of items in list
    LDR    r2, =list       ;load starting address of list
    LDR    r3, =0x9ABCDEF1    ;load key
    MOV    r4, #0;index = 0
loop
    CMP    r4, r1   ;index < n
    BGT    done
    LDR    r5, [r2, r4, LSL #2]     ;load table item at index
    CMP    r5, r3          ;any match with the key?
    BEQ    found ;found, r0 points to matched item
    ADD    r4, r4, #1    ;inc index
    B          loop
found
    ADD    r0, r2, r4, LSL #2;r0 = address of matched item
done  B      done
```

Example 9.4 Sequential Search – ARM codes 2

```
    AREA DATA1, DATA, READWRITE
list
    DCD   0x12345678 ;1st item
    DCD   0x56789ABC    ;2nd item
    DCD   0x9ABCDEF1    ;3rd item
    DCD   0xDEF01234 ;4th item
    DCD   0x00001234 ;5th item
    END
```

Example 9.4 Sequential Search – Demo



Example 9.4 Sequential Search – initial file





Summary

- Representing table/list in memory
- Accessing a table element → base + offset addressing
- Populating a table with data
  - Arithmetic progression
- Jump table → table with addresses of subroutines
- Sort and search a table with key and data
  - Bubble sort
  - Sequential search